AD-A157 916 THE THEORY AND PRACTICE OF A MULTISTATE REFLECTOMETER (U) ROYAL SIGNALS AND RADAR ESTABLISHMENT MALVERN (ENGLAND) J P IDE ET AL. JAN 85 RSRE-MEMO-3824 1/1

UNCLASSIFIED DRIC-BR-96368 F/G 14/2 NL

END
DATE
FILMED
10 85


1·0
2·8
2·5
3·15
2·2
3·5
1·1
4·0
2·0
4·5
1·8
1·25
1·4
1·6

ROYAL SIGNALS AND RADAR ESTABLISHMENT

Memorandum 3824

TITLE: THE THEORY AND PRACTICE OF A MULTISTATE REFLECTOMETER

AUTHORS: J P Ide and L C Oldfield

DATE: January 1985

SUMMARY

This report details the way a multistate reflectometer is used to measure the reflection coefficient of waveguide devices. It contains theoretical derivations of the equations and a computer program, with explanatory notes, for solving them. The calibration procedure is also given. These techniques are suitable for all frequency bands and transmission media where suitable standards and detectors exist.

THE THEORY AND PRACTICE OF A MULTISTATE REFLECTOMETER

J P Ide and L C Oldfield

LIST OF CONTENTS

LIST OF FIGURES

1 INTRODUCTION

The theory of multistate reflectometers has been well documented[1,2]. This paper demonstrates the practical implementation of that theory. The circuit arrangement is a recent development suggested by G F Engen, and has the advantage over earlier designs of a wide range of calibration techniques. The theory of this circuit is given and some of the implications discussed.

The rest of the paper shows the equipment and computer program used to construct a multistate reflectometer for measuring the voltage reflection coefficients of waveguide devices. The program for operating the reflectometer and performing numerical operations on the data is entitled "MULTISTATE" version 2 and has been written in Basic 4.0 for use with the 4000 series Commodore PET.

2 MULTISTATE REFLECTOMETER THEORY

Figure 1 illustrates the multistate reflectometer.



Figure 1

Assume that each position k of the reflector in Figure 1 produces a different repeatable state of linear transmission and reflection within the waveguide junction having ports 1 to 4. Postulating that at any port there is no evanescent mode and only one transmitted mode (but not necessarily the same transmitted mode at all ports), enables each linear state to be described by a scattering matrix $[s]_k$. Then, if a_i and b_i respectively represent waves incident on and emergent from port i (where i = 1,2,3,4); $[b] = [s]_k[a]$. Denoting the reflection coefficients of the detectors providing power indications P_3 and P_4 by $\Gamma_3 = a_3/b_3$ and $\Gamma_4 = a_4/b_4$, and substituting gives:

$$\begin{bmatrix} b_1 \\ b_2 \\ 0 \\ 0 \end{bmatrix} = \begin{bmatrix} s_{11} & s_{12} & s_{13}\Gamma_3 & s_{14}\Gamma_4 \\ s_{21} & s_{22} & s_{23}\Gamma_3 & s_{24}\Gamma_4 \\ s_{31} & s_{32} & (s_{33}\Gamma_3-1) & s_{34}\Gamma_4 \\ s_{41} & s_{42} & s_{43}\Gamma_3 & (s_{44}\Gamma_4-1) \end{bmatrix} \begin{bmatrix} a_1 \\ a_2 \\ b_3 \\ b_4 \end{bmatrix} \qquad (2.1)$$

Since there is non-identical transmission between the parts 1 to 4, the square matrix may be inverted for each k. Denoting the inverse by [M] and its elements by m_{ij}:

$$\begin{bmatrix} a_1 \\ a_2 \\ b_3 \\ b_4 \end{bmatrix} = [M]_k \begin{bmatrix} b_1 \\ b_2 \\ 0 \\ 0 \end{bmatrix} \qquad (2.2)$$

Thus for the multistate 4-port junction 1,2,3,4 for each state k (suppressing the suffix k for the moment)

$$\begin{aligned} a_i &= m_{i1}b_1 + m_{i2}b_2 \qquad \text{(for } i = 1,2\text{)} \\ b_i &= m_{i1}b_1 + m_{i2}b_2 \qquad \text{(where, now, } i = 3,4\text{)} \end{aligned} \qquad (2.3)$$

whence

$$b_i = \frac{m_{i1}}{m_{21}} a_2 + \left(m_{i2} - \frac{m_{22}m_{i1}}{m_{21}}\right) b_2 \quad \text{(for } i = 3,4\text{)} \qquad (2.4)$$

Substituting $\Gamma = a_2/b_2$ gives

$$b_i = (\alpha_i\Gamma + \beta_i)b_2 \qquad \text{(for } i = 3,4\text{)} \qquad (2.5)$$

where $\alpha_i = m_{i1}/m_{21}$ and $\beta_i = (m_{i2} - m_{22}m_{i1}/m_{21})$

Reintroducing the suffix k gives

$$(b_4)_k = (\alpha_4\Gamma + \beta_4)_k \; b_2 \tag{2.6}$$

Similar analysis of the <u>invariant</u> 3-port junction 2,3,5 produces:

$$b_3 = (\alpha_3\Gamma + \beta_3) \; b_2 \tag{2.7}$$

where, now, α_3 and β_3 are dependent on only the scattering coefficients of the 3-port junction and the detector at port 3. Combining equation (2.7) and (2.6) gives

$$\left(\frac{P_4}{P_3}\right)_k = \left|\frac{(b_4)_k}{b_3}\right|^2 = \left|\frac{d_k\Gamma + e_k}{c\Gamma + 1}\right|^2 \tag{2.8}$$

Equation (2.8) also describes a reflectometer with (k+1) detectors connected to a time-invariant junction having (k+3) ports. For $k = 3$ the six-to-four port reduction algorithm can be used, making calibration with three precisely known loads applicable[3,4].

Those methods of calibration were not chosen in this case as speed of calibration was not a criterion and the extra redundancy in the method chosen gives a better assessment of the uncertainties involved in calibration and measurement.

Substituting $\rho e^{j\theta} = \Gamma$ and converting from complex to polar notation equation (2.8) becomes:

$$\left(\frac{P_4}{P_3}\right)_k = L_k\left(\frac{1 + (\rho R_k)^2 + 2\rho R_k \cos(P_k+\theta)}{1 + (\rho S)^2 + 2\rho \, S \cos(F+\theta)}\right) \tag{2.9}$$

It is useful from the viewpoint of calculating uncertainties to assume that S and F are not invariant but are dependent upon the state thus:

$$\left(\frac{P_4}{P_3}\right)_k = L_k\left(\frac{1 + (\rho R_k)^2 + 2\rho R_k \cos(P_k+\theta)}{1 + (\rho S_k)^2 + 2\rho S_k \cos(F_k+\theta)}\right) \tag{2.10}$$

This is a non-linear equation with 5 unknown L,R,P,S,F for each state k. A solution could be found immediately using an iterative least-squares fitting method. But a more accurate method, if low values of reflection coefficient are to be measured, is to determine the k values of L and then use the fitting technique to solve for each R,P,S and F.

2.1 FINDING VALUES OF L

With a perfect load on the output port $\rho = 0$ and suppressing the suffix k again, equation (2.10) becomes $p_4/p_3 = L$. Finding a perfect load is difficult so one is emulated using a fairly good load and precision spacers.

If ρ is small and the directivity of the second coupler is good (giving $S < .05$) then the term (ρS) becomes negligible giving

$$\frac{p_4}{p_3} = L\,(1 + (\rho R)^2 + 2\rho R\cos(P+\theta)) \qquad (2.11)$$

Then introducing known phase shifts α and β by inserting, in turn, two precision spacers between the output port and the load gives 3 equations to solve:

$$\left(\frac{p_4}{p_3}\right)_1 = L\,(1 + (\rho R)^2 + 2\rho R\cos(P+\theta)) \qquad (2.12a)$$

$$\left(\frac{p_4}{p_3}\right)_2 = L\,(1 + (\rho R)^2 + 2\rho R\cos(P+\theta+\alpha)) \qquad (2.12b)$$

$$\left(\frac{p_4}{p_3}\right)_3 = L\,(1 + (\rho R)^2 + 2\rho R\cos(P+\theta+\beta)) \qquad (2.12c)$$

Let $s_1 = (p_4/p_3)_1$, $s_2 = (p_4/p_3)_2$ and $s_3 = (p_4/p_3)_3$ and combine P and θ into a single phase shift θ', then let

$$c_1 = \frac{s_1 - s_2}{s_1 - s_3} = \frac{\cos\theta' - \cos(\theta'+\alpha)}{\cos\theta' - \cos(\theta'+\beta)} \tag{2.13}$$

$$c_1(1-\cos\beta+\tan\theta'\sin\beta) = 1 - \cos\alpha + \tan\theta'\sin\alpha \tag{2.14}$$

$$\tan\theta' = \frac{c_1(1-\cos\beta) + \cos\alpha - 1}{\sin\alpha - c_1(\sin\beta)} \tag{2.15}$$

From this the value of θ' is found. If we then let $c_2 = s_1/s_3$ then

$$c_2 = \frac{1 + (\rho R)^2 + 2\rho R\cos\theta'}{1 + (\rho R)^2 + 2\rho R\cos(\theta'+\beta)} \tag{2.16}$$

Let $c_3 = c_2 \cos(\theta'+\beta) - \cos\theta'$ then

$$(c_2-1)(\rho R)^2 + 2c_3\rho R + c_2-1 = 0 \tag{2.17}$$

This is a simple quadratic in ρR and by putting $c_4 = 2c_3/(c_2-1)$ the required solution is

$$\rho R = \frac{-c_4 + \sqrt{c_4^2 - 4}}{2} \tag{2.18}$$

knowing θ' and ρR then L is calculated from equation (2.12a).

2.2 FINDING VALUES OF R,P,S AND F

Having determined L a movable short is used to evaluate the remaining 4 parameters. Suitable redundancy for the iterative least squares fitting technique is provided by using 7 data points for each state. To minimise the number of iterations required to reach the optimum solution a good set of starting estimates are required. These are called the starting parameters and a new set are required for each state.

2.2.1 Finding Starting Parameters

This part of the calibration procedure is carried out using a short circuit with various phase shifts supplied using spacers (or a movable short circuit) so ρ is replaced by -1. Let $Y = p_4/p_3$ for a particular phase shift X, then equation (2.10) becomes

$$Y = L\left(\frac{1 + R^2 - 2R\cos(P+X)}{1 + S^2 - 2S\cos(F+X)}\right) \tag{2.19}$$

A good set of starting parameters can be found by treating equation (2.19) as a linear equation, ie taking R^2 and R as independent parameters.

Let $a = 1+R^2$, $b = 1+S^2$, $c = -2R\cos P$,
and $d = -2R\sin P$, $e = -2S\cos F$ and $f = -2S\sin F$

Then equation (2.19) becomes

$$\frac{Y}{L} = \left(\frac{a + c\cos X - d\sin X}{b + e\cos X - f\sin X}\right) \tag{2.20}$$

$$\frac{Y}{L} = s1 + s2\cos X - s3\sin X - s4\frac{Y}{L}\cos X + s5\frac{Y}{L}\sin X \tag{2.21}$$

where $s1 = a/b$, $s2 = c/b$, $s3 = d/b$, $s4 = e/b$ and $s5 = f/b$.

A fifth order simultaneous equation solver (using elimination and having no safeguards against the matrix being singular or making any attempt to optimise the sequence of elimination) is used to evaluate s1, s2 etc from equation (2.21). From the original substitutions

$$P = \tan^{-1}(s3/s2) \quad \text{and} \quad F = \tan^{-1}(s5/s4)$$

$$(R/S)^2 = \frac{s2^2+s3^2}{s4^2+s5^2} \quad \text{and} \quad s1 = \frac{1 + R^2}{1 + S^2} \tag{2.22}$$

Then

$$R = \sqrt{\frac{s1 - 1}{1 - s1\left(\frac{s4^2+s5^2}{s2^2+s3^2}\right)}} \quad \text{and} \quad S = \sqrt{\frac{1 + R^2 - s1}{s1}} \qquad (2.23)$$

2.2.2 Least Squares Fitting Procedure

The least squares fitting is done by taking a Taylor expansion about each of the parameters thus linearising the equation to enable normal equations to be written that can be solved to obtain corrections to the original estimates of the parameters. To implement this method the partial derivatives with respect to each of the parameters are required.

From equation (2.19)

$$\frac{\partial Y}{\partial R} = \frac{2L(R-\cos(P+X))}{1+S^2-2S\cos(F+X)} \qquad (2.24a)$$

$$\frac{\partial Y}{\partial P} = \frac{2LR\sin(P+X)}{1+S^2-2S\cos(F+X)} \qquad (2.24b)$$

$$\frac{\partial Y}{\partial S} = \frac{-2Y(S-\cos(F+X))}{1+S^2-2S\cos(F+X)} \qquad (2.24c)$$

$$\frac{\partial Y}{\partial F} = \frac{-2YS\sin(F+X)}{1+S^2-2S\cos(F+X)} \qquad (2.24d)$$

In terms of the estimates of the parameters (making allowance for them being incorrect) we can write

$$Y = L\left[\frac{1+R^2-2R\cos(P+X)}{1+S^2-2S\cos(F+X)}\right] + \left(\frac{\partial Y}{\partial R}\right)\delta R + \left(\frac{\partial Y}{\partial P}\right)\delta P + \left(\frac{\partial Y}{\partial S}\right)\delta S + \left(\frac{\partial Y}{\partial F}\right)\delta F \qquad (2.25)$$

Let $$m = Y - L\left[\frac{1 + R^2 - 2R\cos(P+X)}{1 + S^2 - 2S\cos(F+X)}\right]$$

Then

$$m = \left(\frac{\partial Y}{\partial R}\right)\delta R + \left(\frac{\partial Y}{\partial P}\right)\delta P + \left(\frac{\partial Y}{\partial S}\right)\delta S + \left(\frac{\partial Y}{\partial F}\right)\delta F \qquad (2.26)$$

For a given estimate of the parameters, m,(∂Y/∂R) etc are all known and ∂R etc are unknown.

The least squares method applied to equation (2.26) proceeds by minimising

$$\sum_{i=1}^{i=NP} e_i^2$$

where NP is the number of points and

$$\sum_{i=1}^{i=NP} e_i^2 = \left[m - \left(\frac{\partial Y}{\partial R}\right)\delta R - \left(\frac{\partial Y}{\partial P}\right)\delta P - \left(\frac{\partial Y}{\partial S}\right)\delta S - \left(\frac{\partial Y}{\partial F}\right)\delta F\right]^2_{\text{Point1}} + \left[m - \ldots\ldots\right]^2_{\text{Point2}} + \left[\ldots\ldots\right. \qquad (2.27)$$

This means we are searching for a stationary (minimum) value of Σe_i^2 given by a particular set of δR, δP, δS, and δF. By taking

$$\frac{\partial \, \Sigma e_i^2}{\partial(\delta R)} = 0, \quad \text{etc} \quad ,$$

we obtain the normal equations:

$$
\begin{aligned}
\left[\left(\frac{\partial Y}{\partial R}\right)^2\right]\delta R + \left[\frac{\partial Y}{\partial R}\cdot\frac{\partial Y}{\partial S}\right]\delta S + \left[\frac{\partial Y}{\partial R}\cdot\frac{\partial Y}{\partial P}\right]\delta P + \left[\frac{\partial Y}{\partial R}\cdot\frac{\partial Y}{\partial F}\right]\delta F &= \left[\frac{\partial Y}{\partial R}\cdot m\right] \\
\left[\frac{\partial Y}{\partial S}\cdot\frac{\partial Y}{\partial R}\right]\delta R + \left[\left(\frac{\partial Y}{\partial S}\right)^2\right]\delta S + \left[\frac{\partial Y}{\partial S}\cdot\frac{\partial Y}{\partial P}\right]\delta P + \left[\frac{\partial Y}{\partial S}\cdot\frac{\partial Y}{\partial F}\right]\delta F &= \left[\frac{\partial Y}{\partial S}\cdot m\right] \\
\left[\frac{\partial Y}{\partial P}\cdot\frac{\partial Y}{\partial R}\right]\delta R + \left[\frac{\partial Y}{\partial P}\cdot\frac{\partial Y}{\partial S}\right]\delta S + \left[\left(\frac{\partial Y}{\partial P}\right)^2\right]\delta P + \left[\frac{\partial Y}{\partial P}\cdot\frac{\partial Y}{\partial F}\right]\delta F &= \left[\frac{\partial Y}{\partial P}\cdot m\right] \\
\left[\frac{\partial Y}{\partial F}\cdot\frac{\partial Y}{\partial R}\right]\delta R + \left[\frac{\partial Y}{\partial F}\cdot\frac{\partial Y}{\partial S}\right]\delta S + \left[\frac{\partial Y}{\partial F}\cdot\frac{\partial Y}{\partial P}\right]\delta P + \left[\left(\frac{\partial Y}{\partial F}\right)^2\right]\delta F &= \left[\frac{\partial Y}{\partial F}\cdot m\right]
\end{aligned}
\qquad (2.28)
$$

where the [] brackets indicate summation over all data points.

The four simultaneous equations are solved for δR, δS, δP and δF using the determinant method, which are applied as corrections to the last estimates of R, S, P and F used. The sum squared error is then checked and if the change from the last cycle is $< 5 \times 10^{-5}$ then the current values of R, S, P and F are accepted as the best estimate.

2.3 STANDARD DEVIATION OF THE INDIVIDUAL PARAMETERS

If RQ is the total RMS error of the points then the standard deviation of the parameters is given by the product of RQ and the ratio of two determinants D(P) and D2[5]. D2 is the 4x4 determinant used in the equation solver and D(P) is a 3x3 determinant created from D2 by removing the relevant row and column. Thus D(R) is D2 with the first row and column removed etc.

This cycle of iterative least squares fitting and then finding the standard deviation of the individual parameters is performed for each state.

2.4 MEASUREMENT OF AN UNKNOWN

Reintroducing the suffix k to denote the state, then

$$m_k = \frac{Y_k}{L_k} = \frac{1 + (\rho R_k)^2 + 2\rho R_k \cos(P_k+\theta)}{1 + (\rho S_k)^2 + 2\rho S_k \cos(F_k+\theta)} \qquad (2.29)$$

where ρ and θ are now the unknowns.

2.4.1 Starting Parameters

We linearise the equation by considering $(\rho R_k)^2$ as independent of ρR_k and $(\rho S_k)^2$ as independent of ρS_k. Thus expanding equation (2.29) for 3 states we obtain

$$\begin{aligned}
m_1-1 &= s1(R_1^2-m_1S_1^2)+s2(R_1\cos(P_1)-m_1S_1\cos(F_1)) + \\
&\qquad s3(m_1S_1\sin(F_1)-R_1\sin(P_1)) \\
m_2-1 &= s1(R_2^2-m_2S_2^2)+s2(R_2\cos(P_2)-m_2S_2\cos(F_2)) + \\
&\qquad s3(m_2S_2\sin(F_2)-R_2\sin(P_2)) \qquad (2.30) \\
m_3-1 &= s1(R_3^2-m_3S_3^2)+s2(R_2\cos(P_3)-m_3S_3\cos(F_3)) + \\
&\qquad s3(m_3S_3\sin(F_3)-R_3\sin(P_3))
\end{aligned}$$

where $s1 = \rho^2$, $s2 = 2\rho\cos\theta$ and $s3 = 2\rho\sin\theta$.

Hence starting values required are $\sqrt{s1}$ and $\tan^{-1} s3/s2$.

The 3 simultaneous equations are solved by the determinant method.

2.4.2 2-Parameter Least Square Fitting

This is done in a similar manner to the calibration (section 2.2.2) but is simpler because only two parameters are involved

$$Y_k = L_k\left(\frac{1+(\rho R_k)^2+2\rho R_k\ \cos(P_k+\theta)}{1+(\rho S_k)^2+2\rho S_k\ \cos(F_k+\theta)}\right) = L_k\left(\frac{N}{D}\right) \qquad (2.31)$$

$$\frac{\partial Y_k}{\partial \rho} = \frac{2}{(D)}\ (L_k R_k[\rho R_k+\cos(P_k+\theta)] - Y_k S_k[\rho S_k+\cos(F_k+\theta)] \qquad (2.32)$$

$$\frac{\partial V_k}{\partial \theta} = \frac{2\rho}{(D)}\ (L_k R_k\ \sin(P_k+\theta)\ -\ Y_k S_k\ \sin(F_k+\theta)) \qquad (2.33)$$

and the normal equations are

$$\begin{aligned} \left[\left(\frac{\partial Y}{\partial P}\right)^2\right]\delta\rho + \left[\frac{\partial Y}{\partial \rho}\cdot\frac{\partial Y}{\partial \theta}\right]\delta\theta &= \left[\frac{\partial Y}{\partial \rho}\cdot m\right] \\ \left[\frac{\partial Y}{\partial \rho}\cdot\frac{\partial Y}{\partial \theta}\right]\delta\rho + \left[\left(\frac{\partial Y}{\partial \theta}\right)^2\right]\delta\theta &= \left[\frac{\partial Y}{\partial \theta}\cdot m\right] \end{aligned} \qquad (2.34)$$

where $m = Y_k - L_k(N/D)$ and the [] brackets indicate summation over the k states.

3 AN EXAMPLE MULTISTATE REFLECTOMETER

Figure 2 shows the general layout and equipment used for this multistate reflectometer. The operation of most of the equipment is implicit in its function but a more detailed look at some of the non-standard components is helpful.



FIGURE 2. An Example Multistate Reflectometer

3.1 THE BENTHAM SMD-3D DRIVER AND STEPPING MOTORS

The motors used have 12 Volts 4 phase windings and give a rotation of 1.8°/step. The driver has a front panel switch giving capability of selecting speeds between 1 step/second and 400 steps/second. This facility is not fully available via the IEEE-488 bus. Only one of two internally settable speeds can be selected. The instructions to the driver must be of the form "XNNNNY", where X is either "D" or "I" denoting direction and Y can be any one of four characters indicating which motor and which preset speed. The "NNNN" signifies any 4 digit number in the range "0000" to "9999", which denotes the number of steps to be taken.

3.1.1 Stepping Motor 1

This is coupled to a waveguide switch via a general drive mechanism. This ensures rapid transitions and good alignment with minimum mechanical shock.

3.1.2 Stepping Motor 2

This is used to control the position of the side-arm short circuit. A 1:1 gearbox with nylon gears is used to reduce noise and allow the micrometer to move in and out without having to mount the motor on a movable base. The micrometer thread is 2 turns/mm giving a resolution of 0.0025 mm/step. The resettability is considerably better than that, when precautions are taken to minimise backlash. The present arrangement gives a maximum speed of adjustment of the short circuit of 1 mm/second. Increasing this speed would degrade the resolution.

3.2 MICROWAVE FREQUENCY STABILISATION

The microwave sources used are Systron-Donner YIG tuned Gunn diode devices. They are voltage tunable across a waveguide band and have a separate coil for frequency modulation. The EIP 578 source locking counter has a COARSE TUNE output giving 0 to +10 V and a FINE LOCK output. If the COARSE TUNE output supplies the tuning voltage of the source and a voltage-to-current converter is connected between the FINE LOCK output and the FM coil, then frequency locking can be readily achieved. This system gives excellent frequency stability and the facility of computer selection of any frequency in the band.

4 DETAILS AND EXPLANATION OF COMPUTER PROGRAM

Lines 100-140 Program name and details.

Lines 200-230 Assign channels for IEEE-488 bus.

Line 240 Send voltmeter instructions for range, triggering and SRQ.

Line 250 Set up arrays. In this version of Basic arrays of less than 10 elements do not need to be declared.

Lines 270-320 Input data, frequency and waveguide band information. Determine relevant waveguide broad dimension. This version covers 3 waveguide bands from 8.2-26.5 GHz.

Line 330 Calculate phase-change/mm for that frequency in that waveguide.

Lines 340-380 Print headings.

Line 390 Decide to perform new calibration or use constants stored on a disk from a previous calibration at the same frequency.

Lines 410-550 USING A PREVIOUS CALIBRATION STORED ON DISK.

Lines 430-460 Generate file name and open channel to disk drive. Ensure file exists and check for any disk errors.

Lines 470-490 Acquire micrometer settings of the side-arm short-circuit for each state and the calibration constants at each setting.

Lines 500-550	Print headings and calibration data.
Lines 570-2300	NEW CALIBRATION.
Lines 590-610	Input required micrometer settings and print them.
Lines 630-920	MEASUREMENT OF L (section 2.1).
Lines 650-680	Input length of spacer inserted between the output port and the attached load and measure P_4/P_3 for each state using subroutine one. Repeat for two further spacer lengths.
Lines 690-820	Use circle fitting technique to find the value of L for each state.
Lines 830-920	Print headings, data and final values of L.
Lines 940-2180	DERIVE VALUES OF R, P, S and F (section 2.2).
Lines 960-980	Input information about the short circuit used for calibration and the number of data points required. Using seven points seems to give good results but any number between 5 and 20 allowed.
Lines 1000-1010	Print headings.
Lines 1020-1120	Measure P_4/P_3 in each state using subroutine one for each position of the calibration short circuit. Reverse the order of the states on alternate cycles, and print accumulated data.
Line 1160	Derive calibration constants for each state in turn.
Lines 1180-1540	FIND STARTING PARAMETERS (section 2.2.1).
Line 1200	Select data related to this state.

Lines 1210-1230	Set up matrix of 5 simultaneous equations. Using equation (2.21) and the first five data points.
Lines 1240-1540	Use determinant method to extract initial values of R, P, S and F.
Lines 1560-2010	FOUR PARAMETER ITERATIVE LEAST SQUARES FITTING (section 2.2.2).
Line 1580	Set number of data points and criterion for successful convergence.
Line 1590	Perform up to nine iterations.
Lines 1620-1730	Find partial derivatives at each data point and sum over all data points (equation (2.28)).
Lines 1740-1910	Find corrections s1, s2, s3, s4 to the current values of the parameters using the determinant method of simultaneous equation solving.
Line 1920	Apply corrections to current values of parameters.
Lines 1930-1970	Calculate the RMS value of the differences between predicted and measured values.
Line 1980	Check if either the absolute value of the RMS deviation or the change in RMS deviation between this iteration and the previous one are within the convergence criterion. If converged drop out of loop to Line 2070.
Lines 1990-2010	Print latest values and go back to 1590 for next iteration.
Lines 2050-2060	Print warning if number of iterations reaches 9 without the convergence criterion being satisfied.
Lines 2070-2100	Ensure R and S are positive, and P and F lie between 0 and 2π.

Lines 2110-2150	Calculate the individual standard deviations of the parameters (section 2.3).
Lines 2160-2170	Store calibration data for this state.
Line 2180	Go back to 1200 until all states calibrated.
Lines 2190-2300	Print headings and values of the calibration constants for each state.
Lines 2310-2320	If required store calibration data and constants on disk, using subroutine eight.
Lines 2340-3100	MEASUREMENT OF UNKNOWN (section 2.4).
Line 2360	Set number of parameters to 2 and the number of data points to the number of states.
Lines 2370-2380	Input identity of unknown and print headings.
Line 2390	Measure P_4/P_3 for each state using subroutine one.
Lines 2410-2620	FIND STARTING PARAMETERS (section 2.4.1).
Lines 2430-2520	Set up matrix of three simultaneous equations using equation (2.30) and the first three data points.
Lines 2560-2600	Solve the simultaneous equations using the determinant method.
Lines 2610-2620	Derive starting values for UM and UP where UM and UP are the unknown magnitude and phase respectively.
Lines 2640-2890	TWO PARAMETER ITERATIVE LEAST SQUARES FITTING (section 2.4.2).
Line 2660	Set number of data points and criterion for successful convergence.
Line 2670	Perform up to nine iterations.

Lines 2680-2760	Find partial derivatives at each data point and sum over all data points (equation (2.34)).
Lines 2800-2810	Solve 2 simultaneous equations and apply corrections to current values of the parameters.
Lines 2820-2850	Calculate the RMS difference between predicted and measured values.
Line 2860	Check convergence. If converged drop out of loop to Line 2920.
Lines 2870-2890	Print current values of parameters and RMS Deviation and return to 2670 for next iteration.
Lines 2900-29100	Print warning if number of iterations reaches 9 without convergence criterion being satisfied.
Lines 2920-2930	Ensure $\lvert\rho\rvert$ of unknown (UM) is positive and phase of unknown (UP) lies between 0 and 2π.
Lines 2940-3080	Print out results.
Lines 3100-3110	Either go to Line 2340 if another measurement is required or else close all channels and program.
Lines 3180-4040	SUBROUTINES.
Lines 3180-3370	SUBROUTINE ONE.
Line 3200	Measure the bridge voltages with no microwave power present using subroutine two.
Line 3210	Store bridge voltages.
Line 3220	Turn microwave power on using subroutine five.

Lines 3230-3290	Measure the bridge voltage with microwave power present. Store these voltages. Advance side-arm short circuit to next position using subroutine six. Continue this loop until bridge voltages measured and stored for every state.
Line 3300	Turn microwave power off using subroutine five.
Lines 3310-3320	Measure bridge voltages in the absence of microwave power using subroutine two and store them.
Lines 3330-3360	Calculate the amount of microwave power indicated by each bridge in each state assuming a linear drift with time between each pair of "power off" readings. Find the ratio of indicated powers for each state and store them.
Line 3370	Return to main program.
Lines 3390-3470	SUBROUTINE TWO.
Line 3410	Open all channels on data scanner.
Line 3420	Determine the number of voltage integrations to be performed.
Lines 3430-3440	Measure the bridge voltage of the forward going power sensor and store it.
Lines 3450-3460	Likewise for the reverse power sensor.
Line 3470	Return to subroutine one.
Lines 3490-3550	SUBROUTINE THREE.
Line 3510	Place voltmeter in averaging mode and wait until voltages have settled.
Line 3520	Trigger voltmeter and wait for SRQ using subroutine four to indicate completion of single measurement. Perform this loop until the required number of measurements have been integrated.

Lines 3530-3540	Collect averaged voltage from voltmeter, turn averaging mode off and empty the internal store.
Line 3550	Return to subroutine two.
Lines 3570-3630	SUBROUTINE FOUR.
Lines 3600-3620	Bit 7 of the byte stored in location 59427 will become 1 when a SRQ has been received.
Line 3630	Return to subroutine three.
Lines 3650-3700	SUBROUTINE FIVE.
Line 3670	K determines the number of steps to be performed by stepping motor 1.
Line 3680	Z determines the direction of steps. The change in status byte acknowledges the receipt of the instruction.
Line 3690	A null instruction is sent which is only acknowledged when the previous instruction has been performed.
Line 3700	Return to subroutine one.
Lines 3720-3780	SUBROUTINE SIX.
Lines 3740-3750	Construct instruction containing the number of steps required to move the side-arm short circuit to the right position for the next state.
Lines 3760-3770	Use the status byte and a null instruction to determine successful receipt and completion of the instruction.
Line 3780	Return to subroutine one.

Lines 3800-3840	SUBROUTINE SEVEN.
Line 3820	Keep adding 2π radians until angle is greater than zero.
Line 3830	Keep subtracting 2π radians until angle is less than 2π.
Line 3840	Return to main program.
Lines 3860-4040	SUBROUTINE EIGHT.
Lines 3880-3890	Construct and display file name that the calibration data will be stored under.
Line 3900	Open channel to disk drive.
Line 3910	Check if this filename has already been used. If not, go to Line 3960.
Lines 3920-3950	Facility for overwriting an existing file or returning to main program.
Line 3960	Check for disk errors.
Lines 3970-4030	Store calibration information and data on disk.
Line 4040	Return to main program.

5 LISTING OF COMPUTER PROGRAM

```
100 rem"          MULTISTATE              Version 2              18/1/85
110 rem
120 rem" A program to control a MULTISTATE REFLECTOMETER and solve the
130 rem" non-linear equations involved in calibration and measurement
140 rem
150 rem"
160 rem"    Channel 21   RF bridge FORWARD power meter
170 rem"    Channel 22   RF bridge REVERSE power meter
180 rem"
190 rem
200 open4,4:open8,4,1:open9,4,2: rem" COMMODORE 4022 printer
210 open 6,6:                    rem" DATRON 1071 voltmeter
220 open 7,7:                    rem" RACAL-DANA 1200 switch controller
230 open 129,30:                 rem" BENTHAM SMD-3B stepping motor drive
240 print#6,"r4t2q1"
250 dim df(20),x(20),y(20),yr(20,5),ro(20,5)
260 n2=4: na=1: nb=4: z=1
270 input"date ";da$:input"frequency in ghz. ";fr
280 input "waveguide size x,j or k";wb$
290 if wb$="x" then a=22.860:rem"    8.2 - 12.4 GHz
300 if wb$="j" then a=15.799:rem"    12.4 - 18  GHz
310 if wb$="k" then a=10.668:rem"    18 - 26.5  GHz
320 if a=0 then print:print"       cannot calibrate that band":end
330 kl=4*pi*sqr((fr↑2/299.7↑2)-(1/(4*a↑2)))
340 print#4,chr$(1);"       "+wb$+ "-band reflectometer"
350 print#4:print#4
360 print#4," frequency = ";fr;" ghz";spc(30);da$:print#4:print#4
370 input"any comments?";c$
380 print#4,c$:print#4:print#4
390 input"new or stored calibration";c$:if c$="new" then 570
400 rem"
410 rem"    USING A PREVIOUS CALIBRATION STORED ON DISK
420 rem"
430 file$="1:"+wb$+str$(fr)+"ghz,seq,read"
440 open2,8,5,file$:             rem" COMMODORE 8050 dual disk drive
450 if ds=62 then print"no calibration data for this frequency":close2:end
460 if ds>20 then print ds$:close2:end
470 input#2,dc$,ms(1),ms(2),ms(3),ms(4)
480 fori=1to4:input#2,r(i),p(i),s(i),f(i),l(i):nexti
490 close2
500 print#4,,"using calibration constants from ";dc$:print#4
510 print#4,"  micrometer ";chr$(1);"  rs    ps    ss    fs    ls":print#4
520 c$="      99       99.99999     9.99999     9.99999     9.99999     9.9999
530 print#9,c$
540 for i=1 to 4: print#8,ms(i),r(i),p(i),s(i),f(i),l(i): print#4
550 next i: print#4: goto 2340
560 rem"
570 rem"    NEW CALIBRATION
580 rem"
590 input "micrometer settings";ms(1),ms(2),ms(3),ms(4)
600 print#4,"          using micrometer settings ";
610 for i=1to3: print#4,ms(i);",";: nexti: print#4,ms(4): print#4
620 rem"
630 rem"    MEASUREMENTS USING A LOAD AND SPACERS
640 rem"
650 for i=1 to 3: input "spacer length (mm)";sp: yr(i,0)=-sp*kl
660 gosub 3180: rem" SUBROUTINE ONE
```

```
670 nz=na:na=nb:nb=nz:z=-z
680 nexti
690 alpha=yr(2,0)-yr(1,0): beta=yr(3,0)-yr(1,0)
700 for j=1 to 4
710 s1=yr(1,j): s2=yr(2,j): s3=yr(3,j)
720 c1=(s1-s2)/(s1-s3)
730 c2=(c1*(1-cos(beta)))-(1-cos(alpha))
740 c3=sin(alpha)-(c1*sin(beta))
750 theta=atn(c2/c3): c2=s1/s3
760 if c3<0 then theta=theta+pi
770 ct=cos(theta)
780 c3=(c2*cos(theta+beta))-ct
790 c4=2*c3/(c2-1)
800 rn=(-c4+sqr(c4*c4-4))/2
810 l(j)=s1/(1+rn*rn+2*rn*ct)
820 nextj
830 print#4
840 a%=abs(alpha/pi*180):a$=str$(a%)
850 b%=abs(beta/pi*180):b$=str$(b%)
860 print#4," pos'n     0 degs  ";a$;" degs ";b$;" degs          ls":print#4
870 c$="    9       9.99999    9.99999    9.99999      9.999999"
880 print#9,c$
890 for j=1 to 4:print#8,j,yr(1,j),yr(2,j),yr(3,j),l(j):nextj
900 print#4:print#4
910 print" values of ls"
920 fori=1to 4:print left$(str$(l(i)),8):nexti
930 rem"
940 rem"   | MEASUREMENTS WITH SHORT CIRCUITS ATTACHED |
950 rem"
960 input "length(mm) of short offset";ol
970 print "number of settings of sliding short":input cp
980 av=2*pi/cp: iv=av/kl: pf=-kl*ol
990 m2$="  short offset"+str$(ol)+" mm"
1000 c$="     length    phase     pos'n 1    pos'n 2    pos'n 3    pos'n 4"
1010 print#4,m2$ :print#4:print#4:print#4:print#4,c$ :print#4
1020 for i=1 to cp
1030 sp=iv*(i-1): c$=left$(str$(sp),6)
1040 print " set sliding short to ";c$;" mm"
1050 x(i)=-sp*kl
1060 gosub 3180: rem" SUBROUTINE ONE
1070 nz=na:na=nb:nb=nz:z=-z
1080 c$="   99.999     999.9     9.99999    9.99999    9.99999    9.99999
1090 print#9,c$
1100 print#8,sp,x(i)*180/pi;
1110 for k=1 to 4:print#8,yr(i,k);:nextk:print#8
1120 nexti
1130 rem"
1140 rem"   | SOLVE CALIBRATION EQUATIONS FOR 4 STATES |
1150 rem"
1160 for ic=1 to 4:print"calibration ";ic
1170 rem"
1180 rem"   | FIND STARTING PARAMETERS |
1190 rem"
1200 for i=1 to cp: y(i)=yr(i,ic): next i: l=l(ic)
1210 for j=1 to 5:x=x(j)+pf: y=y(j)
1220 m(j,1)=cos(x):m(j,2)=-sin(x):m(j,3)=-y*cos(x)/l:m(j,4)=y*sin(x)/l
1230 m(j,5)=y/l:next j
1240 for j=1 to 3:j1=j+1:j2=j+2
1250 ma(j)=m(j1,1)-m(j2,1)
1260 mb(j)=m(j2,1)-m(1,1)
```

```
1270 mc(j)=m(1,1)-m(j1,1)
1280 dv(j)=m(1,2)*ma(j)+m(j1,2)*mb(j)+m(j2,2)*mc(j)
1290 next j
1300 a1=(m(1,3)*ma(1)+m(2,3)*mb(1)+m(3,3)*mc(1))/dv(1)
1310 a2=(m(1,3)*ma(1)+m(2,4)*mb(1)+m(3,4)*mc(1))/dv(1)
1320 a3=(m(1,3)*ma(1)+m(2,5)*mb(1)+m(3,5)*mc(1))/dv(1)
1330 a4=(m(1,3)*ma(2)+m(3,3)*mb(2)+m(4,3)*mc(2))/dv(2)
1340 a5=(m(1,4)*ma(2)+m(3,4)*mb(2)+m(4,4)*mc(2))/dv(2)
1350 a6=(m(1,5)*ma(2)+m(3,5)*mb(2)+m(4,5)*mc(2))/dv(2)
1360 a7=(m(1,3)*ma(3)+m(4,3)*mb(3)+m(5,3)*mc(3))/dv(3)
1370 a8=(m(1,4)*ma(3)+m(4,4)*mb(3)+m(5,4)*mc(3))/dv(3)
1380 a9=(m(1,5)*ma(3)+m(4,5)*mb(3)+m(5,5)*mc(3))/dv(3)
1390 t1=a3-a6: t2=a1-a4: t3=a2-a5
1400 t4=a1-a7: t5=a3-a9: t6=a2-a8
1410 s5=(t1*t4-t5*t2)/(t3*t4-t6*t2)
1420 s4=t1/t2-s5*t3/t2
1430 s3=a3-s4*a1-s5*a2
1440 s2=m(1,5)-m(2,5)-s3*(m(1,2)-m(2,2))-s4*(m(1,3)-m(2,3))
1450 s2=(s2-s5*(m(1,4)-m(2,4)))/(m(1,1)-m(2,1))
1460 s1=m(1,5)-s2*m(1,1)-s3*m(1,2)-s4*m(1,3)-s5*m(1,4)
1470 p=atn(s3/s2):if s2>0 then p=p+pi
1480 n0=p: gosub 3800: p=n0
1490 f=atn(s5/s4):if s4>0 then f=f+pi
1500 n0=f: gosub 3800: f=n0
1510 k1=(s4↑2+s5↑2)/(s2↑2+s3↑2)
1520 r=sqr(abs((s1-1)/(1-s1*k1)))
1530 s=sqr((1+r↑2-s1)/s1)
1540 print r,p,s,f: print
1550 rem"
1560 rem"    | 4-PARAMETER LEAST SQUARES FITTING |
1570 rem"
1580 n3=cp:n4=n2-1:c0=5e-5 :rem convergence
1590 for kc=1 to 9
1600 p1=0:q1=0:q2=0:r1=0:r2=0:r3=0:s1=0:s2=0:s3=0:s4=0
1610 f1=0:f2=0:f3=0:f4=0:
1620 for k=1 to n3: xb=x(k)+pf
1630 cu=cos(p+xb):cl=cos(f+xb)
1640 tu=1+r*r-2*r*cu: tl=1+s*s-2*s*cl: tm=2*l/tl
1650 yr=l*tu/tl: dm=y(k)-yr
1660 dm=y(k)-yr:d1=tm*(r-cu):d2=tm*r*sin(p+xb)
1670 d3=-2*yr/tl*(s-cl):d4=-2*yr*s/tl*sin(f+xb)
1680 p1=p1+d1*d1
1690 q1=q1+d1*d2: q2=q2+d2*d2
1700 r1=r1+d1*d3: r2=r2+d2*d3: r3=r3+d3*d3
1710 s1=s1+d1*d4: s2=s2+d2*d4: s3=s3+d3*d4: s4=s4+d4*d4
1720 f1=f1+d1*dm: f2=f2+d2*dm: f3=f3+d3*dm: f4=f4+d4*dm
1730 next k
1740 rem"
1750 rem"    | SOLVE 4 SIMULTANEOUS EQUATIONS |
1760 rem"
1770 t1=r3*s4-s3*s3: t2=r2*s4-s3*s2: t3=r2*s3-r3*s2
1780 t4=r1*s4-s3*s1: t5=r1*s3-r3*s1: t6=r1*s2-r2*s1
1790 u1=r2*s4-s2*s3: u2=q2*s4-s2*s2: u3=q2*s3-r2*s2
1800 u4=q1*s4-s2*s1: u5=q1*s3-r2*s1: u6=q1*s2-q2*s1
1810 v1=r2*s3-s2*r3: v2=q2*s3-s2*r2: v3=q2*r3-r2*r2
1820 v4=q1*s3-s2*r1: v5=q1*r3-r2*r1: v6=q1*r2-q2*r1
1830 ta=q2*t1-r2*t2+s2*t3: tb=q1*t1-r2*t4+s2*t5
1840 tc=q1*t2-q2*t4+s2*t6: td=q1*t3-q2*t5+r2*t6
1850 te=q1*t1-r1*t2+s1*t3: tf=p1*t1-r1*t4+s1*t5
1860 tg=p1*t2-q1*t4+s1*t6: th=p1*t3-q1*t5+r1*t6
```

```
1870 d2=p1*ta-q1*tb+r1*tc-s1*td
1880 s1= (f1*ta-f2*te+f3*(q1*u1-r1*u2+s1*u3)-f4*(q1*v1-r1*v2+s1*v3))/d2
1890 s2=(-f1*tb+f2*tf-f3*(p1*u1-r1*u4+s1*u5)+f4*(p1*v1-r1*v4+s1*v5))/d2
1900 s3= (f1*tc-f2*tg+f3*(p1*u2-q1*u4+s1*u6)-f4*(p1*v2-q1*v4+s1*v6))/d2
1910 s4=(-f1*td+f2*th-f3*(p1*u3-q1*u5+r1*u6)+f4*(p1*v3-q1*v5+r1*v6))/d2
1920 r=r+s1: p=p+s2: s=s+s3: f=f+s4
1930 rq=0
1940 for i=1 to n3: aq=x(i)+pf
1950 ur=1+r*r-2*r*cos(p+aq): lr=1+s*s-2*s*cos(f+aq)
1960 df(i)=y(i)-l*ur/lr: rq=rq+df(i)↑2: next i
1970 rq=sqr(rq/(n3-n2))
1980 if abs(rq)<c0 or abs(r0-rq)<c0 then 2070
1990 print:fori=1ton2: print str$(p3$(i));:nexti:print
2000 print"r.m.s. deviation = ";str$(rq)
2010 r0=rq: next kc
2020 rem"
2030 rem"   | END OF ITERATIONS |
2040 rem"
2050 print "                warning !!!!"
2060 print"  convergence criterion not satisfied "
2070 if r<0 then r=-r: p=p+pi
2080 n0=p: gosub 3800: p=n0
2090 if s<0 then s=-s: f=f+pi
2100 n0=f: gosub 3800: f=n0
2110 d(1)=q2*(r3*s4-s3*s3)-r2*(r2*s4-s2*s3)+s2*(r2*s3-s2*r3)
2120 d(2)=p1*(r3*s4-s3*s3)-r1*(r1*s4-s1*s3)+s1*(r1*s3-s1*r3)
2130 d(3)=p1*(q2*s4-s2*s2)-q1*(q1*s4-s1*s2)+s1*(q1*s2-q1*s2)
2140 d(4)=p1*(q2*r3-r2*r2)-q1*(q1*r3-r1*r2)+r1*(q1*r2-r1*q2)
2150 for i1=1 to 4:b(i1,ic)=sqr(abs(d(i1)/d2))*rq: next i1
2160 for i=1 to n3: r0(i,ic)=df(i): next i: r1(ic)=rq
2170 r(ic)=r:p(ic)=p:s(ic)=s:f(ic)=f
2180 next ic
2190 print#4:print#4
2200 print#4,chr$(1);"  rs    ps    ss    fs    rms    ls"
2210 print#9," 99.99999   9.99999   9.99999   9.99999   .999999   9.99999"
2220 print#4
2230 fori=1to4:print#8,r(i),p(i),s(i),f(i),r1(i),l(i)
2240 print#8,b(1,i),b(2,i),b(3,i),b(4,i):print#4:nexti
2250 print#4,"  residuals":print#4
2260 c$="  phase ":for i=1 to 4:c$=c$+"    pos'n"+str$(i):nexti
2270 print#4,c$:print#4
2280 print#9,"  9.9999   sz.99999   sz.99999   sz.99999   sz.99999
2290 for i=1 to cp:print#8,x(i);
2300 for j=1 to 4:print#8,r0(i,j);:nextj:print#8:nexti:print#4
2310 input "save calibration data on disk";c$:if c$<>"yes" then 2340
2320 gosub 3860: rem" SUBROUTINE EIGHT
2330 rem"
2340 rem"   | END OF CALIBRATION, START OF MEASUREMENT |
2350 rem"
2360 n2=2:cp=4
2370 print"type and ser.no of unknown":inputm2$
2380 print#4,chr$(13);chr$(1);"  measurement of ";m2$:print#4
2390 gosub 3180: rem" SUBROUTINE ONE
2400 rem"
2410 rem"   | STARTING PARAMETERS |
2420 rem"
2430 y1=y(1)/l(1): y2=y(2)/l(2): y3=y(3)/l(3): f1=y1-1: f2=y2-1: f3=y3-1
2440 p1=r(1)↑2-y1*s(1)↑2
2450 p2=r(2)↑2-y2*s(2)↑2
2460 p3=r(3)↑2-y3*s(3)↑2
```

```
2470 q1=r(1)*cos(p(1))-y1*s(1)*cos(f(1))
2480 q2=r(2)*cos(p(2))-y2*s(2)*cos(f(2))
2490 q3=r(3)*cos(p(3))-y3*s(3)*cos(f(3))
2500 r1=y1*s(1)*sin(f(1))-r(1)*sin(p(1))
2510 r2=y2*s(2)*sin(f(2))-r(2)*sin(p(2))
2520 r3=y3*s(3)*sin(f(3))-r(3)*sin(p(3))
2530 rem"
2540 rem"   SOLVE 3 SIMULTANEOUS EQUATIONS
2550 rem"
2560 t1=q2*r3-r2*q3: t2=p2*r3-r2*p3: t3=p2*q3-q2*p3
2570 dv=p1*t1-q1*t2+r1*t3
2580 s1=( f1*t1-f2*(q1*r3-r1*q3)+f3*(q1*r2-r1*q2))/dv
2590 s2=(-f1*t2+f2*(p1*r3-r1*p3)-f3*(p1*r2-r1*p2))/dv
2600 s3=( f1*t3-f2*(p1*q3-q1*p3)+f3*(p1*q2-q1*p2))/dv
2610 um=sqr(abs(s1)):up=atn(s3/s2):if s2<0 then up=up+pi
2620 n0=up: gosub 3800: up=n0: print um,up
2630 rem"
2640 rem"   2-PARAMETER LEAST SQUARES FITTING
2650 rem"
2660 n3=4:n4=1:c0=5e-5 :rem convergence
2670 for kc=1 to 9
2680 p1=0:q1=0:q2=0:f1=0:f2=0
2690 for k=1 to n3: xa=p(k):xb=f(k):rk=r(k)*um:sk=s(k)*up
2700 cu=cos(up+xa):cl=cos(up+xb)
2710 tu=1+rk*rk+2*rk*cu:tl=1+sk*sk+2*sk*cl
2720 yr=l(k)*tu/tl: dm=y(k)-yr
2730 d1=2/tl*(l(k)*r(k)*(rk+cu)-yr*s(k)*(sk+cl))
2740 d2=-2*um/tl*(l(k)*r(k)*sin(up+xa)-yr*s(k)*sin(up+xb))
2750 p1=p1+d1*d1: q1=q1+d1*d2: q2=q2+d2*d2: f1=f1+d1*dm: f2=f2+d2*dm:
2760 next k
2770 rem"
2780 rem"   SOLVE 2 SIMULTANEOUS EQUATIONS
2790 rem"
2800 p2=q1:dv=p1*q2-q1*p2
2810 um=um+(f1*q2-f2*q1)/dv: up=up+(f2*p1-f1*p2)/dv
2820 rq=0:for i=1 to n3: pu=um*r(i): pl=um*s(i)
2830 ur=(1+pu*pu+2*pu*cos(up+p(i)))/(1+pl*pl+2*pl*cos(up+f(i)))
2840 df(i)=y(i)-l(i)*ur/lr: rq=rq+df(i)↑2: next i
2850 rq=sqr(rq/(n3-n2))
2860 if abs(rq)<c0 or abs(r0-rq)<c0 then 2920
2870 print:print um,up: print
2880 print"r.m.s. deviation = ";str$(rq)
2890 r0=rq: next kc
2900 print "                  warning !!!!"
2910 print"  convergence criterion not satisfied "
2920 if um<0 then um=-um: up=up+pi
2930 n0=up: gosub 3800: up=n0
2940 rem"
2950 rem"   MEASUREMENT PRINTOUT
2960 rem"
2970 b1=sqr(abs(q2/dv))*rq: b2=sqr(abs(p1/dv))*rq
2980 c$="   measured       estimate      residual   "
2990 c$=c$+"        mag s.d   phase s.d"
3000 print#4,c$
3010 for i=1 to 4:if i>1 then 3040
3020 c$="  99.99999      99.99999      s.99999             .999999    .999999"
3030 print#9,c$: print#8,y(i),y(i)-df(i),df(i),b1,b2: goto 3060
3040 c$="  99.99999      99.99999      s.99999"
3050 print#9,c$: print#8,y(i),y(i)-df(i),df(i)
3060 nexti:print#4
```

```
3070 r$=str$(um): p$=str$(up*180/pi)
3080 print#4:print#4,"rho of ";m2$;" = ";r$;" at ";p$;" degrees"
3090 nz=na:na=nb:nb=nz:z=-z
3100 input"another device";c$:if c$="yes" then 2340
3110 close4: close6: close7: close8: close9: close129
3120 end
3130 rem"
3140 rem"    END OF PROGRAM
3150 rem"
3160 rem
3170 rem"
3180 rem"    SUBROUTINE ONE
3190 rem"
3200 w2=1:gosub 3390:rem" SUBROUTINE TWO
3210 z1=v1: z2=v2: w2=0
3220 z$="i":gosub 3650 :rem" SUBROUTINE FIVE
3230 for j=na to nb step z:gosub 3390:rem" SUBROUTINE TWO
3240 v1(j)=v1: v2(j)=v2
3250 if j=nb then 3290
3260 ms=ms(j+z)-ms(j)
3270 z$="d": if ms<0 then z$="i"
3280 gosub 3720: rem" SUBROUTINE SIX
3290 nextj
3300 z$="d":gosub 3650 :rem" SUBROUTINE FIVE
3310 w2=1:gosub 3390:rem" SUBROUTINE TWO
3320 z4=v1: z5=v2: w2=0
3330 ac=1:for j=na to nb step z:bc=1-ac
3340 za=(ac*z1)+(bc*z4):zb=(ac*z2)+(bc*z5)
3350 y(j)=(zb↑2-v2(j)↑2)/(za↑2-v1(j)↑2):y$=left$(str$(y(j)),7)
3360 printy$:ac=ac-(1/3):yr(i,j)=y(j):nextj
3370 return
3380 rem"
3390 rem"    SUBROUTINE TWO
3400 rem"
3410 print#7,"rx"
3420 nd=8:if w2=1 then nd=4
3430 print#7,"c21x":gosub 3490:rem" SUBROUTINE THREE
3440 v1=val(v$)
3450 print#7,"c22x":gosub 3490:rem" SUBROUTINE THREE
3460 v2=val(v$)
3470 return
3480 rem"
3490 rem"    SUBROUTINE THREE
3500 rem"
3510 print#6,"a1":foraa=1to100:nextaa
3520 for k=1 to nd:print#6,"@":gosub 3570: rem" SUBROUTINE FOUR
3530 nextk:input#6,v$,w$
3540 print#6,"a010":print#7,"rx"
3550 return
3560 rem"
3570 rem"    SUBROUTINE FOUR
3580 rem"
3590 a=peek(59426)
3600 for zx=1 to 5000
3610 b=peek(59427) and 128:if b=0 then next zx
3620 if zx>4998 then print"srq not detected"
3630 return
3640 rem"
3650 rem"    SUBROUTINE FIVE
3660 rem"
```

```
3670 k$="0200w"
3680 print#129,z$+k$ :if st<>0 then 3680
3690 print#129,"i0000v":if st<>0 then 3690
3700 return
3710 rem"
3720 rem"   SUBROUTINE SIX
3730 rem"
3740 k%=420*ms: k=len(str$(k%)): k$=mid$(str$(k%),2,k-1)
3750 if len(k$)<4 then k$="0"+k$: goto 3750
3760 print#129,z$+k$+"v":if st<>0 then 3760
3770 print#129,"i0000v":if st<>0 then 3770
3780 return
3790 rem"
3800 rem"   SUBROUTINE SEVEN
3810 rem"
3820 if n0<0 then n0=n0+2*pi :goto 3820
3830 if n0>2*pi then n0=n0-2*pi :goto 3830
3840 return
3850 rem"
3860 rem"   SUBROUTINE EIGHT
3870 rem"
3880 cr$=chr$(13):file$="1:"+wb$+str$(fr)+"ghz"
3890 print"file name will be ";file$:print
3900 open2,8,5,file$+",seq,write"
3910 if ds<>63 then 3960
3920 print"    a file of this name already exists"
3930 input"do you wish to over-write it";c$
3940 if c$="no" then print#2,">":close2:return
3950 print#2,">":file$="@"+file$:close2:goto 3900
3960 if ds>20 then printds$:close2:return
3970 m1=ms(1): m2=ms(2): m3=ms(3): m4=ms(4)
3980 print#2,da$;cr$;m1;cr$;m2;cr$;m3;cr$;m4;cr$;
3990 fori=1to4
4000 d1=r(i):d2=p(i):d3=s(i):d4=f(i):d5=l(i)
4010 print#2,d1;cr$;d2;cr$;d3;cr$;d4;cr$;d5;cr$;
4020 nexti:close2
4030 print:print" file ";file$;" written to disk"
4040 return
ready.
```

6 CONCLUSION

The equipment described is very versatile, easy to use and easily adapted to any waveguide band. It produces excellent results at a relatively low cost. The accuracy is dependent upon the accuracy of the standards used to calibrate the reflectometer and the repeatability of the side-arm short circuit. The repeatability can be very tightly controlled if a rotating short circuit is used. The use of a non-contacting short circuit degrades the performance slightly. Other, faster, methods of solving the equations can easily be applied and if an electronic method of changing the state repeatability could be found this would radically improve the speed and usefulness of the equipment. The principle can be applied to media other than waveguide, eg coaxial or stripline, where suitable standards can be found.

ACKNOWLEDGEMENTS

The authors would like to thank E J Griffin for the use of his derivation of equation (2.8) in section 2.

REFERENCES

1 Griffin, E.J., Ide, J.P. and Oldfield, L.C.; "A Multistate Reflectometer", IEEE Trans on Instrum & Meas, Jun 1985.

2 Griffin, E.J.; "The Multistate Reflectometer", RSRE Memo No 3625, Nov 1983.

3 Engen, G.F.; "Calibration of the Six-Port Reflectometer by means of Sliding Terminations", IEEE Trans MTT-26, pp 951-957, Dec 1978.

4 Hodgetts, T.E. and Griffin, E.J.; "A Unified Treatment of the Theory of Calibrating the Six-Port Reflectometer with a Minimum of Standards", RSRE Report No 83003, Aug 1983.

5 Pugh, E.M. and Winslow, G.H.; "The Analysis of Physical Measurements", p 124, Addison-Wesley 1966.

DOCUMENT CONTROL SHEET

Overall security classification of sheet ...UNCLASSIFIED...

(As far as possible this sheet should contain only unclassified information. If it is necessary to enter classified information, the box concerned must be marked to indicate the classification eg (R) (C) or (S))

1. DRIC Reference (if known)	2. Originator's Reference	3. Agency Reference	4. Report Security Classification
	Memorandum 3824		Unclassified

5. Originator's Code (if known)	6. Originator (Corporate Author) Name and Location
	Royal Signals and Radar Establishment
5a. Sponsoring Agency's Code (if known)	6a. Sponsoring Agency (Contract Authority) Name and Location

7. Title

THE THEORY AND PRACTICE OF A MULTISTATE REFLECTOMETER

7a. Title in Foreign Language (in the case of translations)

7b. Presented at (for conference papers) Title, place and date of conference

8. Author 1 Surname, initials	9(a) Author 2	9(b) Authors 3,4...	10. Date	pp. ref.
Ide J P	Oldfield L C			

11. Contract Number	12. Period	13. Project	14. Other Reference

15. Distribution statement

Unlimited

Descriptors (or keywords)

continue on separate piece of paper

Abstract

This Memorandum details the way of a multistate reflectometer is used to measure the reflection coefficient of waveguide devices. It contains theoretical derivations of the equations and a computer program, with explanatory notes, for solving them. The calibration procedure is also given. These techniques are suitable for all frequency bands and transmission media where suitable standards and detectors exist.

S80/48

Line 2660 Set number of data points and criterion for successful convergence.

Line 2670 Perform up to nine iterations.

ATE
LMED
-8